Quantifier Elimination with *Mathematica* #1

ー東大入試を　*1行で*　解くー

2018 年4月8日　Cabri研究会　生越　茂樹

QEのコマンド

f∨g	f || g （or）
f∧g	f && g （and）
¬f	! f（not）
f → g	Implies[f,g]（ =>）
f ⇔ g	Equivalent[f,g]（equiv）
∃x, f	Exists[x,f]
∀x, f	ForAll[x,f]
x∈A	Element[x,A] (elem)
pのQE	Resolve[p,Reals] または Reduce[p,Reals]

- 通常は Resolve の代わりに Reduce でも大丈夫(内部で Resolve をcall)．しかし時間が掛かる.
- ∨, ∧, ⇒ などは“Esc”を使って挟むと入力できる. 例えば “Esc”と”=“と”>”と”Esc” で⇒となる.
- QEはCADを応用している. CAD(Cylindrical Algebraic Decomposition)とは, 空間を符号が一定な部分集合に分割すること.　なお CADはQE以外にも有効で *Mathematica* のコマンドは「GenericCylindricalDecomposition」と「CylindricalDecomposition」.（後者が数学的な厳密解)
- CADの順序の変更が可能. 例えば「Resolve[p,{y,x},Reals]」とすると x□g(y) の形になる. 「Reduce [p,{y,x},Reals]」より速いことが多い.
- 結果の図示は RegionPlot や RegionPlot3D を使う. 測度は RegionMeasure が大体使える.

1．最大・最小

example1&example1’

例1　座標平面上の点 (x,y) が次の方程式を満たす.

$$2x^2+4xy+3y^2+4x+5y-4=0$$

このとき x のとりうる最大値を求めよ.(2012年文科)



例1'　座標平面上の点 (x,y) が次の方程式を満たす.

$$2x^2+4xy+3y^2+4x+5y-4=0$$

このとき $(x+y)$ のとりうる最大値を求めよ.

```
Exists[{y}, 2*x^2 + 4*x*y + 3*y^2 + 4*x + 5*y - 4 == 0];
Resolve[%, Reals]
```

$$\frac{1}{4}\left(-2-5\sqrt{6}\right) \le x \le \frac{1}{4}\left(-2+5\sqrt{6}\right)$$

```
Exists[{x, y}, 2 x^2 + 4 x y + 3 y^2 + 4 x + 5 y - 4 == 0 && x + y == k];
Reduce[%, Reals]
```

$$\frac{1}{4}\left(-4-5\sqrt{2}\right) \le k \le \frac{1}{4}\left(-4+5\sqrt{2}\right)$$

example2

例2

全ての正の実数 x , y に対し

$$\sqrt{x}+\sqrt{y} \le k\sqrt{2x+y}$$

が成り立つような実数 k の最小値を求めよ.(1995年文理)

```
ForAll[{x, y}, x > 0 && y > 0 ⇒ Sqrt[x] + Sqrt[y] ≤ k Sqrt[2 x + y]];
Resolve[%, Reals]
```

$$k \ge \sqrt{\frac{3}{2}}$$

example3

例3

xy 平面内の領域 $-1 \le x \le 1,\ -1 \le y \le 1$ において

$$1-ax-by-axy$$

の最小値が正となるような点(a,b)の範囲を図示せよ.(2000年文)

```
ForAll[{x, y}, -1 ≤ x ≤ 1 && -1 ≤ y ≤ 1 ⇒ 1 - a x - b y - a x y > 0];
Reduce[%, Reals]
```

```
(-1 < a ≤ 0 && -1 < b < 1 + 2 a) || (0 < a < 1 && -1 < b < 1 - 2 a)
```

```
RegionPlot[%, {a, -2, 2}, {b, -2, 2}, Axes → True]
Manipulate[Plot3D[1 - a x - b y - a x y, {x, -1, 1}, {y, -1, 1}], {a, -2, 2}, {b, -2, 2}]
```





example4 (パラメータを含む場合)

例4

a を正の実数とする．次の 2 つの不等式を同時に満たす点 (x, y) 全体からなる領域を D とする．

$$y \geq x^2$$

$$y \leq -2x^2 + 3ax + 6a^2$$

領域D における $x+y$ の最大値，最小値を求めよ.(2004年文)




```
Exists[{x, y}, a > 0 && y ≥ x^2 && y ≤ -2 x^2 + 3 a x + 6 a^2 && k == x + y];
Reduce[%, Reals]
```

$$\left(0 < a \leq \frac{1}{5} \,\&\&\, -a + a^2 \leq k \leq 2a + 4a^2\right) \,||\, \left(\frac{1}{5} < a \leq \frac{1}{2} \,\&\&\, -a + a^2 \leq k \leq \frac{1}{8}\left(1 + 6a + 57a^2\right)\right) \,||\, \left(a > \frac{1}{2} \,\&\&\, -\frac{1}{4} \leq k \leq \frac{1}{8}\left(1 + 6a + 57a^2\right)\right)$$

example5 (パラメーターが二つ以上あるときは結果が非常に複雑)

例5

a, b を実数とする．実数 x, y が $x^2+y^2 \leq 25$, $2x+y \leq 5$ をともに満たすとき

$$z = x^2 + y^2 - 2ax - 2by$$

の最小値を求めよ.(2013年文)



```
Clear[a, b]
Exists[{x, y}, x^2 + y^2 ≤ 25 && 2 x + y ≤ 5 && k == x^2 + y^2 - 2 a x - 2 b y];
Resolve[%, Reals]
```

… 1 …

大きい出力 | 表示を少なく | もっと表示 | すべて表示 | 大きさ制限の設定...

● **MinValue の方が時間がかかる(約10分) が，やや綺麗に出力される.**

```
MinValue[{x^2 + y^2 - 2 a x - 2 b y, x^2 + y^2 ≤ 25 && 2 x + y ≤ 5}, {x, y}] // Timing
```

$$\left\{556.534,\ \begin{cases} -5\,(-5+2\,b) & \cdots 1 \cdots \\ 25-8\,a+6\,b & \cdots 1 \cdots \\ \frac{1}{5}\left(-71+6\,b-4\,b^2\right) & a == 4\ \&\&\ -3 \le b \le 7 \\ \frac{1}{5}\left(25-20\,a-a^2-10\,b+4\,a\,b-4\,b^2\right) & \cdots 1 \cdots \\ -a^2-b^2 & \cdots 1 \cdots \\ -5\left(-5+2\sqrt{16+b^2}\right) & a == 4\ \&\&\ b < -3 \\ -5\left(-5+2\sqrt{a^2+b^2}\right) & \text{True} \end{cases}\right\}$$

大きい出力 | 表示を少なく | もっと表示 | すべて表示 | 大きさ制限の設定...

2. 通過領域

example6

例6 正の実数 a に対して，座標平面上で次の放物線を考える.

$$C:\ y = ax^2 + \frac{1-4a^2}{4a}$$

a が正の実数全体を動くとき，C の通過する領域を求めよ.(2015年理)

```
Exists[a, a > 0 && y == a * x^2 + (1 - 4 a^2) / (4 a)];
reg = Reduce[%, Reals]
```

$(y \le 0\ \&\&\ -1 < x < 1)\ ||\ \left(y > 0\ \&\&\ -\sqrt{1+y^2} \le x \le \sqrt{1+y^2}\right)$

```
RegionPlot[reg, {x, -5, 5}, {y, -5, 5}, Axes → True]
```



```
Manipulate[
 Plot[{a * x^2 + (1 - 4 a^2) / (4 a), Sqrt[x^2 - 1], -Sqrt[x^2 - 1]}, {x, -5, 5}, PlotRange → {{-5, 5}, {-20, 20}}], {a, 0.1, 30}]
```



3. 真偽判定

example7 (整数問題だが解ける)

例7

以下の命題A, B それぞれに対し，その真偽を調べよ．
また真ならば証明を与え，偽ならば反例を与えよ．

命題A　n が正の整数ならば，$\dfrac{n^3}{26}+100 \geq n^2$が成り立つ．

命題B　整数 n,m,l が $5n+5m+3l=1$ をみたすならば，
$10nm+3ml+3nl<0$ が成り立つ．　(2015年文)

```
ForAll[n, n > 0 ⇒ n^3 / 26 + 100 ≥ n^2];
Resolve[%, Integers]
ForAll[{n, m, l}, 5 n + 5 m + 3 l == 1 ⇒ 10 n m + 3 m l + 3 n l < 0];
Resolve[%, Integers]
```

False

True

example8 (整数問題は通常解けない)

例8

n を正の整数, a を実数 とする.　全ての整数 m に対して

$$m^2-(a-1)m+\frac{n^2}{2n+1}a>0$$

が成り立つような a の範囲を n を用いて表せ. (1997年理)

```
ForAll[m, m^2 - (a - 1) m + n^2 / (2 n + 1) a > 0 && a ∈ Reals && n > 0];
Reduce[%, Integers]
```

$$\text{Reduce}\left[\forall_m \left(-(-1+a)\,m+m^2+\frac{a\,n^2}{1+2\,n}>0\;\&\&\;a\in\text{Reals}\;\&\&\;n>0\right),\ \text{Integers}\right]$$

4. 総合問題

example9 (最大・最小)

例9 平面上に点O を中心とする半径 1 の円C がある．
また,この平面上の O と異なる点 A を通って直線
OA と垂直な空間直線 l があり, 平面とのなす角
が 45°である．このとき円C と直線 l の間の
最短距離を, 2点O,A 間の距離 a で表せ.
(1983年理)



C上の点を(x,y,0), l上の点を(a,0,0)+t(0,1,1)=(a,t,t) とおく

```
Exists[{x, y, t}, (a - x)^2 + (t - y)^2 + t^2 == k^2 && x^2 + y^2 == 1 && a ≥ 0 && k ≥ 0];
reg = Reduce[%, {a, k}, Reals]
RegionPlot[reg, {a, 0, 3}, {k, 0, 2}]
```

$\left(0 \le a \le \frac{1}{2} \,\&\&\, k \ge \frac{\sqrt{1-2a^2}}{\sqrt{2}}\right) \,||\, \left(\frac{1}{2} < a \le 1 \,\&\&\, k \ge 1-a\right) \,||\, (a > 1 \,\&\&\, k \ge -1+a)$



example10 (最大・最小)

例10 0 以上の実数 s,t が $s^2+t^2=1$ を満たしながら動くとき,方程式

$$x^4-2(s+t)x^2+(s-t)^2=0$$

の解のとる範囲を求めよ．(2005年文)

```
Exists[{s, t}, s^2 + t^2 == 1, x^4 - 2 (s + t) x^2 + (s - t)^2 == 0];
Resolve[%, Reals]
```

$-2^{3/4} \le x \le 2^{3/4}$

example11 (通過領域)

座標空間内を,長さ 2 の線分 AB が次の 2 条件 (a),(b) を満たしながら動く．
(a) 点Aは 平面 $z=0$ 上にある.
(b) 点C(0,0,1) が線分AB上にある.
このとき線分ABが通過することのできる範囲を K とする．K と
不等式 $z \ge 1$ の表す範囲との共通部分の体積を求めよ. (2016年理)



A(p,q,0),B(x,y,z)とおくと,C(0,0,1)が線分AB上より $\overrightarrow{AB}=t\overrightarrow{AC} \Leftrightarrow \begin{pmatrix} x-p \\ y-q \\ z \end{pmatrix} = t\begin{pmatrix} -p \\ -q \\ 1 \end{pmatrix}$ (t>1).

```
ClearAll["Global`*"]
```

```
Exists[{p, q, t}, {x - p, y - q, z} == t {-p, -q, 1} && z > 1 && (x - p)^2 + (y - q)^2 + z^2 == 4];
surface = Resolve[%, Reals]
```

$z > 1 \,\&\&\, 8z - 3z^2 + x^2 z^2 + y^2 z^2 - 2z^3 + z^4 == 4$

```
(*上の式からx²+y²をzの式で表すと，x^2+y^2=(4-8z+3z^2+2z^3-z^4)/z^2 となるから体積と概形は*)
v = Integrate[Pi (4 - 8 z + 3 z^2 + 2 z^3 - z^4) / z^2, {z, 1, 2}]
```

$\pi\left(\frac{17}{3} - 8\,\mathrm{Log}[2]\right)$

```
ContourPlot3D[-4 + 8 z - 3 z² + x² z² + y² z² - 2 z³ + z⁴ == 0, {x, -1/2, 1/2}, {y, -1/2, 1/2}, {z, 1, 2}]
```



もっと自動的にやってみる．領域を作り，RegionMeasure で計算する．変数の順序に注意!

線分CB上の点をR (X, Y, Z) とすると，$\overrightarrow{CR} = u\,\overrightarrow{CB} \Leftrightarrow \begin{pmatrix} X \\ Y \\ Z-1 \end{pmatrix} = u \begin{pmatrix} x \\ y \\ z-1 \end{pmatrix}$ $(0 < u < 1)$ だから，

```
Clear[x, y, z, u];
Exists[{x, y, z, u}, {X, Y, Z - 1} == u {x, y, z - 1} && 0 < u < 1 && surface];
pred = Resolve[%, Reals] // FullSimplify
region = ImplicitRegion[pred, {Z, X, Y}]; (*{X,Y,Z}では計算できない．{Z,Y,X}はOK*)
RegionMeasure[region]
```

$Z > 1 \,\&\&\, \left(\left(X^2 + Y^2 == 3(-1+Z)^2 \,\&\&\, X^2 + Y^2 + (-1+Z)^2 == 0\right) \,||\, \left(X^2 + Y^2 < 3(-1+Z)^2 \,\&\&\, Z\left(8 + Z\left(X^2 + Y^2 + (-3+Z)(1+Z)\right)\right) < 4\right)\right)$

$2\pi\left(\frac{17}{6} - 4\,\mathrm{Log}[2]\right)$

最後に

■最大·最小値だけなら Maximize/Minimize, MaxValue/MinValue.
また変数の値が1つに決まるときはSolve を使った方が良いことが多い.

■一般にはQEは実数かつ多項式の系以外には使えないとされている．しかしMathematicaでは 使える場合もある.
例えば複素数の場合は Resolve[%, Complexes] と, a∈Reals, (a|b) ∈Reals などを組み合わせると，使える事が多い.

(i)「 x>0 のとき $\left(1+\frac{1}{x}\right)^x < e < \left(1+\frac{1}{x}\right)^{x+\frac{1}{2}}$ を示せ 」 (2016年理)

```
ForAll[x, x > 0 ⇒ (1 + 1 / x) ^x < E < (1 + 1 / x) ^ (x + 1 / 2)]
Resolve[%, Reals]
```

$\forall_x \left(x > 0 \Rightarrow \left(1 + \frac{1}{x}\right)^x < e < \left(1 + \frac{1}{x}\right)^{\frac{1}{2}+x}\right)$

True

(ii)「 x^2+(2m+3i)x+(8-6i)=0 が実数解を持つような 実定数m の値を定めよ．」

```
Exists[x, x^2 + (2 m + 3 I) x + (8 - 6 I) == 0 && (m | x) ∈ Reals]
Resolve[%, Complexes]
```

$\exists_x \left((8 - 6\,i) + x^2 + x\,(3\,i + 2\,m) == 0 \,\&\&\, (m \mid x) \in \mathrm{Reals}\right)$

$m \in \mathrm{Reals} \,\&\&\, 12 + 4\,m == 0$

■工学ではQEは「最適化問題」によく応用されている.
入試問題を解く道具としては，特に「最大·最小問題」,「図形と式」,「方程式·不等式」の分野に有効.
しかし「確率」ではまったく使えない.「整数」でもごく一部しか使えない.
このように万能ではないが，QEは「可視化」,「計算(特に微積分)」にならぶ強力な武器になる．(3種の神器?)

5. (おまけ)これらも一行で解ける

1.【方程式の解】正三角形PQRが放物線$y = x^2$に内接し, QRの傾きが$\sqrt{2}$の時, 1辺の長さを求めよ. (2004 年)
これは Solve でも解ける.

$P\left(p, p^2\right)$, $Q\left(q, q^2\right)$, $R\left(r, r^2\right)$ とおくと QRの傾きは $q + r$. さらに1辺の長さを a とおく.

```
Exists[{p, q, r}, p ≠ q && q ≠ r && r ≠ p &&
   (p - q) ^2 + (p^2 - q^2) ^2 == (q - r) ^2 + (q^2 - r^2) ^2 == (r - p) ^2 + (r^2 - p^2) ^2 == a^2 && r + q == Sqrt[2] && a > 0] // Resolve
```

$$a == \frac{18}{5}$$

2.【最大･最小】$x^3 - 2x^2 - 3x + 4$ $\left(-\frac{7}{4} \leqq x \leqq 3\right)$ の最大･最小値を求めよ. (1991 年)

```
Exists[x, x^3 - 2 x^2 - 3 x + 4 == k && -7 / 4 ≤ x ≤ 3];
Resolve[%, Reals]
```

$$-\frac{143}{64} \le k \le \frac{2}{27}\left(19 + 13\sqrt{13}\right)$$

3.【軌跡】$z = y^2$, $x = 0$ 上に動点Qをとる. さらに定点P (2, 0, 1) としたとき, 直線PQ と xy平面との交点Rの軌跡を求めよ.(1991 年)

$Q\left(0, t, t^2\right)$, $R\ (x, y, 0)$ とおくと $\overrightarrow{PR} = s\,\overrightarrow{PQ}$

```
Exists[{t, s}, {x - 2, y, -1} == s {-2, t, t^2 - 1}];
curve = Resolve[%, Reals]
RegionPlot[ImplicitRegion[%, {x, y}], Axes → True]
```

$$1 - \frac{x}{2} \ne 0 \,\&\&\, -\frac{x}{2} + \frac{x^2}{4} - y^2 == 0$$



4.【連立方程式が特定の解を持つ条件】$y = k\,(x - x^3)$, $x = k\,(y - y^3)$ が第1象限内でy = x上にない交点を持つ正の数kの範囲を求めよ.(1989 年)

```
Exists[{x, y}, y == k (x - x^3) && x == k (y - y^3) && x ≠ y && k > 0 && x > 0 && y > 0]
Resolve[%, Reals]
```

$$k > 2$$

5.【円と放物線】円: $x^2 + (y - a)^2 = r^2$(a, r>0) が $y \geq x^2$ に含まれるとき, rの範囲をaの式で表せ. (有名問題)

```
ForAll[{x, y}, x^2 + (y - a) ^2 ≤ r^2 ⇒ y ≥ x^2 && a > 0 && r > 0];
Resolve[%, {a, r}, Reals]
```

$$\left(0 < a \le \frac{1}{2} \,\&\&\, 0 < r \le a\right) \,||\, \left(a > \frac{1}{2} \,\&\&\, 0 < r \le \frac{1}{2}\sqrt{-1 + 4a}\right)$$

6.【最小値】長さl ($l \geqq 1$) の線分が その両端を$y = x^2$ の上に置いている時, 線分の中点Mがx軸に最も近い場合のMのy座標を求めよ.(1974 文理 改)

```
Exists[{p, q}, (p - q) ^2 + (p^2 - q^2) ^2 == l^2 && y == (p^2 + q^2) / 2 && l ≥ 1] // Resolve[#, Reals] &
```

$$l \ge 1 \,\&\&\, y \ge \frac{1}{4}\,(-1 + 2\,l)$$

7.【存在条件】$f(x) = ax^3 + bx^2 + cx + d$ が 3条件
(イ) f (-1) = 0,　(ロ) f (1) = 0,　(ハ)$|x| \leqq 1$の時 $f(x) \geqq 1 - |x|$
をみたすのは, 定数a, b, c, dがどのような条件をみたすときか？ (1981 理)

```
Clear[f, x, a, b, c, d]; f[x_] = a x^3 + b x^2 + c x + d;
ForAll[x, (Abs[x] ≤ 1 ⇒ f[x] ≥ 1 - Abs[x]) && f[1] == f[-1] == 0] // Reduce[#, Reals] &
```

$$b \le -1 \,\&\&\, \frac{1}{2}\,(1 + 2b) \le a \le \frac{1}{2}\,(-1 - 2b) \,\&\&\, d == -b \,\&\&\, c == -a$$

8.【通過領域】A$\left(\frac{2\left(t^2+t+1\right)}{3\,(t+1)}, -2\right)$, B$\left(\frac{2}{3}t, -2t\right)$ と定める. t が $0 \leqq t \leqq 1$ を動くとき 直線AB の通過領域を求めよ. (1997 文)

```
(*AB:y=-2t^3+3xt^2-3x となるので*)
Exists[t, y == -2 t^3 + 3 x t^2 - 3 x && 0 ≤ t ≤ 1] // Resolve[#, Reals] & // FullSimplify
RegionPlot[%, {x, -3, 4}, {y, -4, 4}]
```

$(2+y \geq 0 \,\&\&\, ((x>0 \,\&\&\, 3x \leq 2 \,\&\&\, 3x+y \leq x^3) \,||\, (x \leq 0 \,\&\&\, 3x+y \leq 0))) \,||$
$(3x+y \geq 0 \,\&\&\, ((3x>2 \,\&\&\, x \leq 1 \,\&\&\, 3x+y \leq x^3) \,||\, (x>1 \,\&\&\, 2+y \leq 0)))$



9. 【通過領域】P $(\frac{1}{2}, \frac{1}{4})$, $y = x^2$上の2点Q (a, a^2), R (b, b^2) を PQ = PR となるように動かすとき △PQRの重心G (X, Y) の軌跡を求めよ. (2011 理)

```
Exists[{a, b}, (a - 1 / 2) ^2 + (a^2 - 1 / 4) ^2 == (b - 1 / 2) ^2 + (b^2 - 1 / 4) ^2 && X == 1 / 3 (a + b + 1 / 2) && Y == 1 / 3 (a^2 + b^2 + 1 / 4)] //
  Resolve[#, Reals] & // FullSimplify
RegionPlot[ImplicitRegion[%, {X, Y}]]
```

$1 + 4X(-1+3X) == 8Y \,||\, (1+4X(-1+3X) < 8Y \,\&\&\, X(2+24Y) == 3+4Y)$



10. 【通過領域】A (-1, 1), B (1, -1), P (x, y) (| x | ≦ 1) に対し次の (1) または (2) をみたすP の存在範囲を求めよ.
(1) 頂点のx座標の絶対値が1以上の2次関数でA, B, P を通るものが存在する.
(2) A, P, B は一直線上にある (2015 年文).

```
Clear[a, p, q, f, x]
f[x_] = a (x - p) ^2 + q;
Exists[{a, p, q}, (1 == f[-1] && -1 == f[1] && Y == f[X] && -1 ≤ X ≤ 1 && Abs[p] ≥ 1) || Y == -X];
Resolve[%, Reals] // FullSimplify
RegionPlot[%, {X, -1, 1}, {Y, -1, 1}, Axes → True]
```

$(-1 \leq X \leq 1 \,\&\&\, ((-2+X)X == 1+2Y \,||\, X(2+X)+2Y == 1 \,||$
$(X+Y \geq 0 \,\&\&\, ((-2+X)X \geq 1+2Y \,||\, X(2+X)+2Y \leq 1)) \,||\, (X+Y \leq 0 \,\&\&\, (X(2+X)+2Y \geq 1 \,||\, (-2+X)X \leq 1+2Y)))) \,||\, X+Y == 0$



11.【条件付最大最小】

3辺の長さが a と b と c の直方体を,
長さが b の 1 辺を回転軸として 90°
回転させるとき, 直方体が通過する
点全体がつくる立体を V とする.
(1) V の体積を a, b, c を用いて表せ.
(2) $a+b+c=1$ のとき, V の体積の
とりえる値の範囲を求めよ. (2010年)



Vは底面が図の斜線部分, 高さがbの柱体になるので volume (V) = $\left(\frac{\pi}{4}(a^2+c^2)+ac\right)b$

```
Exists[{a, b, c}, a + b + c == 1 && a > 0 && b > 0 && c > 0 && (Pi (a^2 + c^2) / 4 + a c) b == k] // Resolve[#, Reals] &
```

$0 < k < \frac{\pi}{27}$

12.【不等式の証明】

$f(x) = x^4 + x^3 + \frac{1}{2}x^2 + \frac{1}{6}x + \frac{1}{24}$, $g(x) = x^5 + x^4 + \frac{1}{2}x^3 + \frac{1}{6}x^2 + \frac{1}{24}x + \frac{1}{120}$
とする．このとき，以下のことが成り立つことを示せ.
(1) 任意の実数 x に対し，$f(x) > 0$ である.
(2) 方程式　$g(x) = 0$ はただひとつの実数解 α をもち, $-1 < \alpha < 0$ となる.(1994年東大理)

```
f[x_] = x^4 + x^3 + 1 / 2 x^2 + 1 / 6 x + 1 / 24;  g[x_] = x^5 + x^4 + 1 / 2 x^3 + 1 / 6 x^2 + 1 / 24 x + 1 / 120;
(*1*) ForAll[x, f[x] > 0] // Resolve[#, Reals] &
```

True

```
(*2*)Exists[α, g[α] == 0 && -1 < α < 0 && ForAll[x, g[x] == 0 ⇒ x == α]] // Resolve[#, Reals] &
```

True